

Chapter 7

Applications of Integration

The first three sections in this chapter use Maple to compute the standard geometric applications of integration: areas, volumes, arc lengths and surface areas. The last section uses Fourier series to approximate functions by sums of multiples of sines and cosines.

7.1 Area

We present two examples in this section: the area between a curve and the x -axis, and the area between two curves.

EXAMPLE 1: Find the area that lies between the graph of

$$f(x) = -0.128x^3 + 1.728x^2 - 5.376x + 2.864$$

and the x -axis.


SOLUTION: Start by inputting an expression for f into Maple.

```
> f:=-0.128*x^3+1.728*x^2-5.376*x+2.864;
```

$$f := -0.128x^3 + 1.728x^2 - 5.376x + 2.864$$

Now plot f over an interval that shows all points where f crosses the x -axis. By trial and error, the interval $-2 \leq x \leq 10$ will do.

```
> plot(f,x=-2..10);
```



The graph of f crosses the x -axis at three points—between 0 and 1, between 3 and 4, and between 9 and 10. Use `fsolve` to find these roots and assign them to variables, say a , b and c .

```
> a,b,c:=fsolve(f=0,x);
      a, b, c := 0.6697777844, 3.631759112, 9.198463104
```

Note that all the roots can be found with `fsolve` without specifying a range for x , because f is a polynomial. From the plot, it is evident that the graph of f is below the x -axis between a and b and above the x -axis between b and c . Therefore, the area is the following sum of integrals:

$$A = - \int_a^b f dx + \int_b^c f dx$$

In Maple we enter

```
> -Int(f,x=a..b)+Int(f,x=b..c); A:=value(%);
      - \int_{0.6697777844}^{3.631759112} -0.128x^3 + 1.728x^2 - 5.376x + 2.864 dx
      + \int_{3.631759112}^{9.198463104} -0.128x^3 + 1.728x^2 - 5.376x + 2.864 dx
      A := 25.05016797
```

As mentioned in the last chapter, the `Int` command (with an uppercase `I`) displays the integral without computing its value so that it can be checked for typing errors. Then the `value` command evaluates it. The area of interest is about 25.05 square units.

EXAMPLE 2: Compute the area between the graph of f (defined above) and the graph of


$$g(x) = \ln x$$

SOLUTION: Once again, we start by defining g as an expression in Maple.

```
> g:=ln(x);
      g := ln(x)
```

Now plot f and g on the same coordinate axes with the command

```
> plot({f,g},x=-2..10);
```



The plot shows three points of intersection, which can again be found with `fsolve`, but this time you need to specify ranges since g is not a polynomial.

```
> a:=fsolve(f=g,x); b:=fsolve(f=g,x=4..5);
> c:=fsolve(f=g,x=8..9);
```

```
a := 0.7587367455
```

```
b := 4.270958448
```

```
c := 8.793410794
```

The graph of f is below the graph of g between a and b while g is below f between b and c . Therefore, the area between the graphs is

$$A = \int_a^b (g - f) dx + \int_b^c (f - g) dx$$

In Maple this is entered as

```
> Int(g-f,x=a..b)+Int(f-g,x=b..c); A:=value(%);
```

$$\int_{0.7587367455}^{4.270958448} \ln(x) + 0.128x^3 - 1.728x^2 + 5.376x - 2.864 dx$$

$$+ \int_{4.270958448}^{8.793410794} -0.128x^3 + 1.728x^2 - 5.376x + 2.864 - \ln(x) dx$$

```
A := 18.17292990
```

7.2 Volume


We look at volume by slicing and volume of revolution.

Volume by Slicing: To calculate a volume by slicing, the first step is to derive an expression that represents the cross sectional area. Then this expression is integrated over the appropriate interval.

EXAMPLE 1: The base of a solid is bounded by the curves $y = -x^2 + 5x - 2$ and $y = x$. The cross sections perpendicular to the x -axis are equilateral triangles. Find the volume of the solid.

SOLUTION: First, define f and g in Maple and graph them

```
> f:=-x^2+5*x-2; g:=x;
      f := -x2 + 5x - 2
      g := x
> plot({f,g},x=0..4);
```



From the plot, notice that f and g cross at two points: one between $x = 0$ and $x = 1$ and the other between $x = 3$ and $x = 4$. Find these roots using `fsolve` and assign them to the variables a and b

```
> a,b:=fsolve(f=g,x);
      a, b := 0.5857864376, 3.414213562
```

The cross section of this solid is an equilateral triangle of side

```
> s:=f-g;
      s := -x2 + 4x - 2
```

The area of an equilateral triangle is $A = \frac{\sqrt{3}s^2}{4}$. So the cross sectional area is

```
> A:=sqrt(3)/4*s^2;
      A :=  $\frac{\sqrt{3}(-x^2 + 4x - 2)^2}{4}$ 
```

The total volume of this solid is obtained by integrating this expression over the interval $a \leq x \leq b$.

```
> Int(A,x=a..b); V:=value(%);
      V := 2.612789059
```

Volume of Revolution: To calculate a volume of revolution, the first step is to determine if it is an x - or y -integral. The second step is to determine if a slice rotates into a disk, washer or cylindrical shell. The third step is to compute the integral with the appropriate formula. Assuming x -integrals, the formulas are:

$$\int_a^b \pi R^2 dx \dots\dots\dots \text{disk}$$

R is the radius of the disk.

$$\int_a^b \pi(R^2 - r^2) dx \dots \text{washer}$$

R and r are the outer and inner radii of the washer.


$$\int_a^b 2\pi rh dx \dots\dots\dots \text{cylinder (shells)}$$

r and h are the radius and height of the cylinder.

EXAMPLE 2: Consider the region that is bounded by the curves $y = -x^2 + 5x - 2$ and $y = x$. Find the volume of the solid that is obtained by revolving this region about the x -axis.

SOLUTION: The region is the same as in Example 1. As in that example, we want to do an x -integral. So we graph the curves with a slice perpendicular to the x -axis.

```
> plot([f,g, [[2.5,subs(x=2.5,f)], [2.5,subs(x=2.5,g)]]],
> x=0..4, thickness = [2,2,5]);
```



Notice that the third quantity plotted is the line segment from $(2.5, f(2.5))$ to $(2.5, g(2.5))$ and it is given thickness 5 by putting the quantities to be plotted and the corresponding parameters in square brackets. When this slice is rotated about the x -axis, it sweeps out a washer. The outer radius is f and the inner radius is g . So the volume is

```
> Int(Pi*(f^2-g^2),x=a..b); V:=value(%);
      \int_{0.5857864376}^{3.414213562} \pi((-x^2 + 5x - 2)^2 - x^2) dx
      V := 66.34705188
```

EXAMPLE 3: Consider the region that is bounded by the curves $y = -x^2 + 5x - 2$ and $y = x$. Find the volume of the solid that is obtained by revolving this region about the y -axis.

SOLUTION: The region is the same as in Examples 1 and 2. Again we want to do an x -integral. The graph of the curves and a slice perpendicular to the x -axis is the same as in Example 2, but when this slice is rotated about the y -axis, it sweeps out a cylindrical shell. The radius is x and the height is $f-g$. So the volume is

$$\begin{aligned} &> \text{Int}(2*\text{Pi}*x*(f-g), x=a..b); \text{ V:=value(%)}; \\ &\int_{0.5857864376}^{3.414213562} 2\pi x(-x^2 + 4x - 2) dx \\ &V := 47.39075134 \end{aligned}$$

7.3 Arc Length and Surface Area


Most textbook problems on arc length and surface area are contrived so that the integrals are doable. Not so in the real world.

Arc Length of a Lissajous Figure: A Lissajous figure is a curve in the plane which may be parametrized as

$$(x, y) = (\cos(pt), \sin(qt))$$

where p and q are positive integers. For example here is a Lissajous figure with $p = 3$ and $q = 4$.

> plot([cos(3*t), sin(4*t), t=0..2*Pi]);



As t varies from 0 to 2π , the x -coordinate oscillates $p = 3$ times while the y -coordinate oscillates $q = 4$ times. (Here, 3 bumps on the left and right and 4 bumps on the top and bottom.)

EXAMPLE 1: Find the arc length of the Lissajous figure with $p = 3$ and $q = 4$.

SOLUTION: The arc length of a parametric curve is given by

$$L = \int_a^b \sqrt{\left(\frac{dx}{dt}\right)^2 + \left(\frac{dy}{dt}\right)^2} dt$$

So define $x(t)$ and $y(t)$ and take their derivatives:

```
> xt,yt:=cos(3*t),sin(4*t);
      xt, yt := cos(3t), sin(4t)
> Dxt:=diff(xt,t); Dyt:=diff(yt,t);
      Dxt := -3 sin(3t)
      Dyt := 4 cos(4t)
```

Then the arclength is

```
> Int(sqrt(Dxt^2+Dyt^2),t=0..2*Pi); L:=value(%);
```

$$\int_0^{2\pi} \sqrt{9 \sin^2(3t) + 16 \cos^2(4t)} dt$$

$$L := \int_0^{2\pi} \sqrt{9 \sin^2(3t) + 16 \cos^2(4t)} dt$$

Oops! Maple cannot do it exactly. (No one can!) So do it numerically:


```
> L:=evalf(%);
      L := 21.23721523
```

Surface Area of Revolution: When a curve $y = f(x)$ is rotated about an axis, the area of the resulting surface is given by

$$A = \int_a^b 2\pi r \sqrt{1 + \left(\frac{dy}{dx}\right)^2} dx$$

where r is the radius of revolution. Specifically, $r = x$ if it is rotated about the y -axis and $r = y = f(x)$ if it is rotated about the x -axis.

EXAMPLE 2: The curve $y = 3 + \cos x$ for $0 \leq x \leq 4\pi$ is revolved about the x -axis. Find the surface area.



SOLUTION: Enter the function and compute its derivative:

```
> y1:=3+cos(x);
```

$$y1 := 3 + \cos(x)$$

```
> Dy1:=diff(y1,x);
```

$$Dy1 := -\sin(x)$$

The curve is rotated about the x -axis; so $r = y$. So the surface area is

```
> Int(2*Pi*y1*sqrt(1+Dy1^2),x=0..4*Pi); A:=value(%);
```

$$\int_0^{4\pi} 2\pi(3 + \cos(x))\sqrt{1 + \sin(x)^2} dx$$


$$A := 48\pi\sqrt{2}\text{EllipticE}\left(\frac{\sqrt{2}}{2}\right) - \pi\ln(\sqrt{2} - 1) + \frac{1}{2}\pi\ln(3 - 2\sqrt{2})$$

The EllipticE function is not very informative. So get a decimal approximation:

```
> A:=evalf(%);
```

$$A := 288.0361274$$

EXAMPLE 3: The curve $y = 3 + \cos x$ for $0 \leq x \leq 4\pi$ is revolved about the y -axis. Find the surface area.



SOLUTION: The curve is the same as in Exercise 2, but it is rotated about the y -axis; so $r = x$. So the surface area is

```
> Int(2*Pi*x*sqrt(1+Dy1^2),x=0..4*Pi); A:=value(%);
```

$$\int_0^{4\pi} 2\pi x\sqrt{1 + \sin(x)^2} dx$$

$$A := \int_0^{4\pi} 2\pi x\sqrt{1 + \sin(x)^2} dx$$

Maple cannot do it exactly. So do it numerically:

```
> A:=evalf(%);
```

$$A := 603.2614546$$

7.4 Introduction to Fourier Series - Cosine Expansions

This is an optional section.

One of the interesting applications of integration is the approximation of functions. We will demonstrate in this section that smooth functions can be

approximated by sums of trigonometric functions called Fourier series. More precisely, we will demonstrate that a function on an interval of the form $[0, L]$ can be approximated by sums of the functions

$$1, \cos\left(\frac{\pi x}{L}\right), \cos\left(\frac{2\pi x}{L}\right), \dots, \cos\left(\frac{n\pi x}{L}\right), \dots \quad (7.1)$$

To do this, we need to learn some properties of the functions above. First, notice that

$$\int_0^L 1 \, dx = L.$$

and if k is a positive integer, then

$$\int_0^L \cos\left(\frac{k\pi x}{L}\right) \, dx = \frac{L}{k\pi} \sin(k\pi) = 0$$

Further, if k is a positive integer, then

$$\int_0^L \cos^2\left(\frac{k\pi x}{L}\right) \, dx = \frac{L}{2} \left(1 + \frac{\sin 2k\pi}{2k\pi}\right) = \frac{L}{2}$$

and if k and m are positive integers with $k \neq m$, then

$$\int_0^L \cos\left(\frac{k\pi x}{L}\right) \cos\left(\frac{m\pi x}{L}\right) \, dx = 0$$

(Do the integration and see for yourself!)

Let's use this information to find a way to approximate a function $f(x)$ in the form

$$f(x) \approx a_0 + a_1 \cos\left(\frac{\pi x}{L}\right) + \dots + a_n \cos\left(\frac{n\pi x}{L}\right) \quad (7.2)$$

on the interval $[0, L]$.

In order to determine the coefficient a_0 , we simply integrate both sides of (7.2) above and obtain

$$\int_0^L f(x) \, dx = a_0 L$$

(since all the other terms integrate to zero). This gives

$$a_0 = \frac{1}{L} \int_0^L f(x) \, dx \quad (7.3)$$

(Note that a_0 is the average value of $f(x)$ on the interval $[0, L]$.)

Next, we obtain a formula for a_1 by multiplying both sides of (7.2) by $\cos\left(\frac{\pi x}{L}\right)$ and integrating. This yields

$$\int_0^L f(x) \cos\left(\frac{\pi x}{L}\right) \, dx = \int_0^L a_1 \cos^2\left(\frac{\pi x}{L}\right) \, dx = a_1 \frac{L}{2}$$

since all the other terms integrate to zero. Consequently, we have

$$a_1 = \frac{2}{L} \int_0^L f(x) \cos\left(\frac{\pi x}{L}\right) dx$$

a_2 can be found in a similar manner by multiplying both sides of (7.2) by $\cos\left(\frac{2\pi x}{L}\right)$ and integrating. In this case we obtain

$$a_2 = \frac{2}{L} \int_0^L f(x) \cos\left(\frac{2\pi x}{L}\right) dx$$

In general, for $k = 1, 2, \dots, n$, we obtain

$$a_k = \frac{2}{L} \int_0^L f(x) \cos\left(\frac{k\pi x}{L}\right) dx \quad (7.4)$$

The numbers a_0, a_1, \dots, a_n are called the **Fourier cosine coefficients** of the function $f(x)$ on the interval $[0, L]$.

DEFINITION: Let $f : [0, L] \rightarrow \mathbb{R}$ be a smooth function. If n is a positive integer then the n^{th} order **Fourier cosine expansion** for $f(x)$ on the interval $[0, L]$ is

$$a_0 + \sum_{k=1}^n a_k \cos\left(\frac{k\pi x}{L}\right)$$

where a_0 and a_k are given by the formulas in (7.3) and (7.4) respectively.

EXAMPLE: Compute the 8th order Fourier cosine expansion for the function $f(x) = x^2 - 1$ on the interval $[0, 3]$. Plot $f(x)$ and its Fourier cosine expansion.

SOLUTION: Using the formulas above, we can input f and compute the Fourier coefficients as follows:

```
> f:=x^2-1;
```

$$f := x^2 - 1$$

```
> 1/3*Int(f,x=0..3); a[0]:=value(%);
```

$$\frac{1}{3} \int_0^3 x^2 - 1 dx$$

$$a_0 := 2$$

```
> 2/3*Int(f*cos(k*Pi*x/3),x=0..3);
```

```
> a[k]:=value(% assuming k::posint;
```

$$\frac{2}{3} \int_0^3 (x^2 - 1) \cos\left(\frac{k\pi x}{3}\right) dx$$

$$a_k := \frac{36(-1)^k}{k^2 \pi^2}$$

Notice the use of the **assume** facility to say that k is a positive integer. Try this command without the assumption. (Read the help pages by typing **?assume** and **?assuming**.)


Consequently, the 8th order Fourier cosine expansion for $f(x) = x^2 - 1$ on the interval $[0, 3]$ is

```
> fcos:=a[0] + sum(a[k]*cos(k*Pi*x/3),k=1..8);
```

$$f_{\text{cos}} := 2 - \frac{36 \cos\left(\frac{\pi x}{3}\right)}{\pi^2} + \frac{9 \cos\left(\frac{2\pi x}{3}\right)}{\pi^2} - \frac{4 \cos(\pi x)}{\pi^2} + \frac{9 \cos\left(\frac{4\pi x}{3}\right)}{\pi^2} - \frac{36 \cos\left(\frac{5\pi x}{3}\right)}{25\pi^2} + \frac{\cos(2\pi x)}{\pi^2} - \frac{36 \cos\left(\frac{7\pi x}{3}\right)}{49\pi^2} + \frac{9 \cos\left(\frac{8\pi x}{3}\right)}{16\pi^2}$$

Finally, plot $f(x)$ and this expression on $[0, 3]$ to see the quality of the approximation.


```
> plot({f, fcos}, x=0..3);
```



It is almost impossible to see a difference between the curve and the approximation. Besides some wiggles, the only deviation is near the endpoint $x = 3$. The reason for the poor behavior there is simple. All of the functions in (7.1) (with $L = 3$) have a zero derivative at $x = 3$. Since the function $f(x) = x^2 - 1$ does not have this property, we can expect the approximating Fourier expansion to have some trouble near $x = 3$. Notice that there is no a problem at $x = 0$ since $f'(0) = 0$, and all of the functions in (7.1) have this property.

What happens when we look at these graphs on a larger interval? First let's try the interval $[-3, 3]$:


```
> plot({f, fcos}, x=-3..3);
```



The agreement is quite good. If you are wondering whether this will always happen, then the answer is a definite NO! The reason it happens here is because $f(x) = x^2 - 1$ is an even function, and all of the functions in (7.1) are also even functions!

Finally, let's look at a larger interval, say $[-4, 10]$.

```
> plot({f, fcos}, x=-4..10, -5..20);
```



Notice that the Fourier cosine expansion falls away quickly outside the interval $[-3, 3]$. You could say that the expansion did what it was asked to do. However, in actuality, the fall off takes place because all of the functions in (7.1) are periodic with period 6 (since $L = 3$) and $f(x)$ is not a periodic function.

7.5 Summary

- Use `fsolve` to find successive points of intersection of the graphs of Maple expressions f and g . Use `Int` and `value` or `Int` and `evalf` syntax to compute integrals.
- **Area:** Integrate the difference $f - g$ or $g - f$ between intersection points to find the area.
- **Volumes by Slicing:** A slice of the base has length given by the difference $f - g$. Use this to find the area of a cross sectional slice of the solid. Integrate this cross sectional area to get the volume.

$$\int_a^b A(x) dx$$

- **Volumes of Revolution:** Determine if you need an x - or y -integral. Determine if a slice rotates into a disk, washer or cylindrical shell. Integrate to get the volume:

$$\int_a^b \pi R^2 dx \dots\dots \text{disk}$$

R is the radius of the disk.

$$\int_a^b \pi(R^2 - r^2) dx \dots\dots \text{washer}$$

R and r are the outer and inner radii of the washer.

$$\int_a^b 2\pi r h dx \dots\dots \text{cylinder (shells)}$$

r and h are the radius and height of the cylinder.

- **Arc Length:** Determine if you need an x -, y - or t -integral. Integrate the arc length differential

$$\begin{aligned} ds &= \sqrt{1 + \left(\frac{dy}{dx}\right)^2} dx \\ &= \sqrt{\left(\frac{dx}{dy}\right)^2 + 1} dy \\ &= \sqrt{\left(\frac{dx}{dt}\right)^2 + \left(\frac{dy}{dt}\right)^2} dt \end{aligned}$$

to get the arc length.

- **Surface Area:** Determine if you need an x -, y - or t -integral. The surface area is the integral

$$\int_a^b 2\pi r ds$$

where ds is the arc length differential and r is the radius of revolution. Specifically, $r = x$ if it is rotated about the y -axis and $r = y$ if it is rotated about the x -axis.

- **Fourier Cosine Series:** Use `Int` and `value` to find the Fourier coefficients. Use `sum` or `Sum` and `value` to create the Fourier cosine series approximations.

7.6 Exercises

1. Find the area of the region that is bounded above by the curves $y = 10\ln(x)$ and $y = 4 - x^4 - x$ and below by the x -axis.
2. Consider the region in the previous exercise.
 - (a) Find the volume of the solid obtained by revolving this region about the x -axis.
 - (b) Find the volume of the solid obtained by revolving this region about the y -axis.

- (c) Find the volume of the solid obtained by revolving this region about the line $x = 4$.
- (d) Find the volume of the solid obtained by revolving this region about the line $y = 4$.
3. The region in the first quadrant bounded by the coordinate axes and the graph of $y = 15 - 2x^2$ is revolved about the x -axis to form a solid. Use `leftsum` and `evalf` to approximate the volume of this solid by summing the volumes of 20, 40, 200, and 500 disks of equal thickness. Is this an overestimate or underestimate of the volume? Repeat with `rightsum`.
4. Compute the volume of the solid in Exercise 3 by integration and compare your answers.
5. *The volume of a JELL-O™ mold.* Consider the region above the x -axis below the parabola that passes through the points $(8, 0)$, $(10, 4)$ and $(12, 0)$.
- (a) Find the volume of the solid obtained by revolving this region about the y -axis.
- (b) Find the volume of the solid obtained by revolving this region about the line $x = 2$.
- (c) Find the surface area obtained by revolving the parabola about the y -axis.
- (d) Find the surface area obtained by revolving the parabola about the line $x = 2$.
6. In this exercise, you are asked to calculate the approximate volume of a bowl of depth 3 feet with circular horizontal cross sections. The measurements of the radius of the cross sections versus height are given in the following table (in feet).

Height	0.00	0.50	1.00	1.50	2.00	2.50	3.00
Radius	0.00	0.55	1.05	1.40	1.70	1.85	2.00

Find an approximate value of the volume of this bowl by adding the volumes of the disks of thickness 0.5 feet and radius given by the values in the table above.

7. Find a third degree polynomial whose graph contains every other data point in the table of the previous exercise. Use this polynomial to compute an approximate volume for the bowl. Compare your answer to that of the previous exercise.

HINT: Think of the x -axis as representing depth, and enter $p := ax^3 + bx^2 + cx + d$ as an arrow-defined function. Solve four equations for the unknowns a , b , c and d so that the graph of p passes through the points $(0, 2)$, $(1, 1.7)$, $(2, 1.05)$, and $(3, 0)$. Graph this polynomial to check it contains these data points.

8. Plot the Lissajous Figure

$$(x, y) = (\cos(pt), \sin(qt))$$

for $p = 3$ and $q = 2$ and find its arc length. Now plot the Lissajous Figure for $p = 6$ and $q = 4$ and find its arc length. Explain what you had to change to get the correct arc length.

9. A circular doughnut is formed by revolving a circle of radius b centered at $x = a, y = 0$ about the y -axis ($a > b$). Your job is to find the formula for the volume of this doughnut and evaluate it when $a = 4$ and $b = 3$.

- In order to have the integral evaluate properly, you will need to communicate to Maple the assumptions that $a > b$ and $b > 0$. Use `assume(a>b,b>0);`
- Find the equation of the circle described above for general a and b .
- Review the cylindrical shell technique for finding volumes. Set up the integral for the volume of this doughnut and use Maple's `Int` and `value` commands to evaluate it. Check it using Pappas' Theorem which says this volume is equal to the product of the area of a circle of radius b and the circumference of a circle of radius a .
- Evaluate the volume in the special case where $a = 4$ and $b = 3$.
- To see a three-dimensional picture of this doughnut, issue the following Maple commands.

```
> x1:=(4+3*cos(t))*cos(s);
> y1:=3*sin(t);
> z1:=(4+3*cos(t))*sin(s);
```

See if you can figure out why these three equations parameterize the doughnut as the parameters s and t vary from 0 to 2π . Then plot using

```
> plot3d([x1,y1,z1], s=0..2*Pi, t=0..2*Pi,
> scaling=constrained);
```

10. In this problem, your job is to find the formula for the surface area of the doughnut in the previous problem and evaluate it when $a = 4$ and $b = 3$.

- Enter the assumptions that $a > b$ and $b > 0$.
- Enter the following parametrization of the circle: $x = 4 + 3 \cos t$, $y = 3 \sin t$. Plot it to check it is a circle.
- Enter the general parametrization: $x = a + b \cos t$, $y = b \sin t$ and compute the surface area of revolution for the doughnut for general a and b . Check it using Pappas' Theorem which says this area is equal to the product of the circumference of a circle of radius b and the circumference of a circle of radius a .
- Evaluate the surface area in the special case where $a = 4$ and $b = 3$.

11. Generate the 5-th and 10-th order Fourier cosine expansions for $\sin(x)$ on the interval $[0, 7]$. Plot each of these against the function $\sin(x)$.
12. Repeat the previous exercise with $f(x) = 6 - x^2$ on the interval $[0, 4]$.
13. In Section 3.2, we plotted polar curves using the `polarplot` command from the `plots` package. We now want to compute the area of the region that lies between a polar curve $r = r(\theta)$ and the origin and between the angles $\theta = a$ and $\theta = b$. This area is given by

$$\frac{1}{2} \int_a^b r(\theta)^2 d\theta$$

Compute the area that is to the left of the parabola $r_1(\theta) = \frac{2}{1 - \cos(\theta)}$ and inside the circle $r_2(\theta) = 8$.

HINT: First plot the two curves to estimate the angles where they cross. Next use `fsolve` to find the angles more precisely. Then perform the appropriate integrals using the `Int` and `evalf` commands.

14. The arc length of a polar curve $r = r(\theta)$ for $a \leq \theta \leq b$, is given by

$$L = \int_a^b \sqrt{r(\theta)^2 + r'(\theta)^2} d\theta$$

and the surface areas of revolution of this polar curve about the x - and y -axes are, respectively

$$A = \int_a^b 2\pi y(\theta) \sqrt{r(\theta)^2 + r'(\theta)^2} d\theta$$

$$A = \int_a^b 2\pi x(\theta) \sqrt{r(\theta)^2 + r'(\theta)^2} d\theta$$

Here, $x(\theta) = r(\theta) \cos(\theta)$ and $y(\theta) = r(\theta) \sin(\theta)$.

- (a) Plot the polar curve $r = 3 + \cos(\theta)$, for $0 \leq \theta \leq \pi/2$.
 - (b) Compute the arc length of the curve $r = 3 + \cos(\theta)$, for $0 \leq \theta \leq \pi/2$.
 - (c) Compute the area of the surface obtained by revolving about the x -axis the curve $r = 3 + \cos(\theta)$, for $0 \leq \theta \leq \pi/2$.
 - (d) Compute the area of the surface obtained by revolving about the y -axis the curve $r = 3 + \cos(\theta)$, for $0 \leq \theta \leq \pi/2$.
15. Plot the rose $r = \cos(5\theta)$. Compute the area of the surface obtained by revolving the petal for $-\pi/10 \leq \theta \leq \pi/10$, first about the x -axis and then about the y -axis. Be careful in choosing your limits of integration.